Mat 1. 2-timersprøve den11. maj 2023

JE 9.5.23

Opgave 1

```
> restart;with(LinearAlgebra):with(plots):
```

En funktion $f : \mathbb{R}^2 \to \mathbb{R}$ er givet ved forskriften:

```
> f:=(x,y)->x^2*y-3*x^2-4*y^2+8*y+11:
> f(x,y)
```

$$x^2 y - 3x^2 - 4y^2 + 8y + 11 \quad \textbf{(1.1)}$$

Spørgsmål 1

$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = (2xy - 6x,\ x^2 - 8y + 8) = (2x(y-3), x^2 - 8y + 8) = (0, 0) \Leftrightarrow$
$2x(y-3) = 0$ og $x^2 - 8y + 8 = 0 \Leftrightarrow x = 0$ og y = 1 eller y = 3 og x = ± 4.
Samtlige stationære punkter for f er da (0, 1), (4, 3) og (−4, 3).

Spøgrsmål 2

Hvis f har lokalt ekstremum i et punkt, så må punktet være et stationært punkt, da f ikke har undtagelsespunkter.

Hessematricen for f i punktet (x, y) er

```
> H(x,y):=<diff(f(x,y),x,x),diff(f(x,y),x,y);diff(f(x,y),y,x),diff(f
  (x,y),y,y)>
```

$$H(x, y) := \begin{bmatrix} 2y - 6 & 2x \\ 2x & -8 \end{bmatrix} \quad \textbf{(1.2.1)}$$

```
> H(0,1):=subs(x=0,y=1,H(x,y))
```

$$H(0, 1) := \begin{bmatrix} -4 & 0 \\ 0 & -8 \end{bmatrix} \quad \textbf{(1.2.2)}$$

```
> Eigenvalues(H(0,1),output=list)
```

$$[-4, -8] \quad \textbf{(1.2.3)}$$

Da begge egenværdierne for **H**(0, 1) er negative har f egentligt lokalt maksimum i det stationære punkt (0, 1).

```
> H(4,3):=subs(x=4,y=3,H(x,y))
```

$$H(4, 3) := \begin{bmatrix} 0 & 8 \\ 8 & -8 \end{bmatrix} \quad \textbf{(1.2.4)}$$

```
> Eigenvalues(H(4,3),output=list)
```

$$[-4 + 4\sqrt{5}, -4 - 4\sqrt{5}] \quad \textbf{(1.2.5)}$$

Da de to egenværdier for **H**(4, 3) har modsat fortegn, så har f hverken lokalt maksimum eller lokalt minimum i det stationære punkt (4, 3) (saddelpunkt).

```
> H(-4,3):=subs(x=-4,y=3,H(x,y))
```

$$H(-4,3) := \begin{bmatrix} 0 & -8 \\ -8 & -8 \end{bmatrix} \quad (1.2.6)$$

```
> Eigenvalues(H(-4,3),output=list)
```

$$\left[-4+4\sqrt{5}, -4-4\sqrt{5}\right] \quad (1.2.7)$$

Da de to egenværdier for **H**(−4, 3) har modsat fortegn, så har f hverken lokalt maksimum eller lokalt minimum i det stationære punkt (−4, 3) (saddelpunkt).

```
> contourplot(f(x,y),x=-6..6,y=-2..5,contours=40)
```



Spørgsmål 3

Parablen med ligningen $y = \frac{1}{4}x^2 - 1$ afgrænser sammen med den rette linie med ligningen $y = 3$ en begrænset og afsluttet punktmængde M.

```
> implicitplot({y=1/4*x^2-1,y=3},x=-4..4,y=-1..3,scaling=
  constrained)
```



Da M er begrænset og afsluttet og da f er kontinuert i M, så har f et globalt minimum og et globalt maksimum i M. Da f ikke har undtagelsespunkter i det indre af M, så antages disse værdier enten i et stationært punkt i det indre af M eller på randen af M.

Det eneste stationære punkt i det indre af M er (0, 1), hvor vi har funktionsværdien

```
> 'f(0,1)'=f(0,1)
```

$$f(0,1) = 15 \tag{1.3.1}$$

Værdien af f på de to randkurver er

```
> simplify(f(x,1/4*x^2-1))
```

$$-1 \tag{1.3.2}$$

for alle $x \in [-4; 4]$.

og

```
> f(x,3)
```

$$-1 \tag{1.3.3}$$

for alle $x \in [-4; 4]$.

Ved numerisk sammenligning af disse undersøgelser fås, at det globale maksimum er 15, som antages i punktet (0, 1) og at det globale minimum er −1, som antages overalt på randen af M.

Det bemærkes, at da M er sammenhængende, så er værdimængden $f(M) = [-1; 15]$.

```
> plot3d(f(x,y),x=-4..4,y=-1..3)
```



Opgave 2

```
> restart;with(LinearAlgebra):with(plots):
```

Et område M i (x, z)-planen er givet ved

$$M = \{(x, z) |\ 0 \le x \le \frac{1}{4}\pi \text{ og } \sin(x) \le z \le \cos(x)\}$$

```
> p0:=plot([cos(x),sin(x)],x=0..Pi/4,scaling=constrained,color=
  black):
> p1:=plot([Pi/4,v,v=0..1],linestyle=dash):
```

```
> display(p0,p1,tickmarks=[2,3],labels=[x,z],title="M")
```



Spørgsmål 1

En parameterfremstilling for *M* er

```
> s:=(u,v)-><u,sin(u)+v*(cos(u)-sin(u))>:
> s(u,v)
```

$$\begin{bmatrix} u \\ \sin(u) + v\,(\cos(u) - \sin(u)) \end{bmatrix} \tag{2.1.1}$$

hvor $u \in [0;\ \frac{\pi}{4}]$ og $v \in [0;\ 1]$ (det lodrette liniestykke fra punktet (u,sin(u)) til punktet (u,cos(u))).

```
> M:=<diff(s(u,v),u)|diff(s(u,v),v)>
```

$$M := \begin{bmatrix} 1 & 0 \\ \cos(u) + v\,(-\sin(u) - \cos(u)) & \cos(u) - \sin(u) \end{bmatrix} \tag{2.1.2}$$

```
> Js:=Determinant(M)
```

$$Js := \cos(u) - \sin(u) \tag{2.1.3}$$

som er ≥ 0, da $u \in [0; \frac{\pi}{4}]$. Den til **s** hørende Jacobi-funktion er da

```
> Jacobi:=Js
```

$$Jacobi := \cos(u) - \sin(u) \tag{2.1.4}$$

$$\mathrm{Ar}(M) = \int_M \mathrm{d}\mu = \int_0^1 \int_0^{\frac{\pi}{4}} \mathrm{Jacobi}(u,v)\mathrm{d}u\mathrm{d}v$$

```
> integranden:=Jacobi
```

$$integranden := \cos(u) - \sin(u) \tag{2.1.5}$$

```
> Int(integranden,[u=0..Pi/4,v=0..1])=int(integranden,[u=0..Pi/4,v=
  0..1])
```

$$\int_0^1 \int_0^{\frac{\pi}{4}} (\cos(u) - \sin(u))\, \mathrm{d}u\, \mathrm{d}v = \sqrt{2} - 1 \tag{2.1.6}$$

Et omdrejningslegeme Ω har parameterfremstillingen

```
> r:=(u,v,w)-><u*cos(w),u*sin(w),sin(u)+v*(cos(u)-sin(u))>
```

$$r := (u, v, w) \mapsto \langle u \cdot \cos(w), u \cdot \sin(w), \sin(u) + v \cdot (\cos(u) - \sin(u)) \rangle \tag{2.1}$$

```
> r(u,v,w)
```

$$\begin{bmatrix} u\cos(w) \\ u\sin(w) \\ \sin(u) + v\,(\cos(u) - \sin(u)) \end{bmatrix} \tag{2.2}$$

hvor $u \in [0; \frac{\pi}{4}]$, $v \in [0; 1]$ og $w \in [0; \pi]$.

Det bemærkes, at Ω netop er fremkommet ved, at området M i (x, z)-planen drejes vinklen π omkring z-aksen mod uret set fra z-aksens positive ende.

Spørgsmål 2

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>
```

$$M := \begin{bmatrix} \cos(w) & 0 & -u\sin(w) \\ \sin(w) & 0 & u\cos(w) \\ \cos(u) + v\,(-\sin(u) - \cos(u)) & \cos(u) - \sin(u) & 0 \end{bmatrix} \tag{2.2.1}$$

```
> Jr:=simplify(Determinant(M))
```

$$Jr := -u\,(\cos(u) - \sin(u)) \tag{2.2.2}$$

som er ≤ 0, da $u \in [0; \frac{\pi}{4}]$. Den til **r** hørende Jacobi-funktion er da

```
> Jacobi:=-Jr
```

$$Jacobi := u\,(\cos(u) - \sin(u)) \tag{2.2.3}$$

$$\mathrm{Vol}(\Omega) = \int_{\Omega} \mathrm{d}\mu = \int_0^{\pi}\int_0^1\int_0^{\frac{\pi}{4}} \mathrm{Jacobi}(u,v,w)\mathrm{d}u\mathrm{d}v\mathrm{d}w$$

```
> integranden:=Jacobi
```

$$integranden := u\,(\cos(u) - \sin(u)) \tag{2.2.4}$$

```
> Int(integranden,[u=0..Pi/4,v=0..1,w=0..Pi])=int(integranden,[u=0..
  Pi/4,v=0..1,w=0..Pi])
```

$$\int_0^{\pi}\int_0^1\int_0^{\frac{\pi}{4}} u\,(\cos(u) - \sin(u))\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w = \left(-1 + \frac{\pi\sqrt{2}}{4}\right)\pi \tag{2.2.5}$$

Opgave 3

```
> restart;with(LinearAlgebra):with(plots):
> vop:=proc(X) op(convert(X,list)) end proc:
```

I (x, y)-planen er der givet hastighedsvektorfeltet $\mathbf{V}(x, y) = \begin{bmatrix} \frac{1}{2}x - \frac{1}{2}y \\ -\frac{1}{2}x + \frac{1}{2}y \end{bmatrix}$.

Spørgsmål 1

Til bestemmelse af flowkurverne for **V** haves differentialligningssystemet

$$\begin{bmatrix} x'(t) \\ y'(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2}x(t) - \frac{1}{2}y(t) \\ -\frac{1}{2}x(t) + \frac{1}{2}y(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{bmatrix} \begin{bmatrix} x(t) \\ y(t) \end{bmatrix},\ t \in \mathbb{R}.$$

```
> A:=<1/2,-1/2;-1/2,1/2>
```

$$A := \begin{bmatrix} \frac{1}{2} & -\frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{bmatrix} \tag{3.1.1}$$

```
> Eigenvectors(A,output=list)
```

$$\left[\left[1, 1, \left\{\begin{bmatrix} -1 \\ 1 \end{bmatrix}\right\}\right], \left[0, 1, \left\{\begin{bmatrix} 1 \\ 1 \end{bmatrix}\right\}\right]\right] \tag{3.1.2}$$

Samtlige flowkurver for **V** er da givet ved

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = c_1 \begin{bmatrix} 1 \\ 1 \end{bmatrix} + c_2 \begin{bmatrix} -1 \\ 1 \end{bmatrix} e^t = \begin{bmatrix} c_1 - c_2 e^t \\ c_1 + c_2 e^t \end{bmatrix}, t \in \mathbb{R}, c_1, c_2 \in \mathbb{R}.$$

Til bestemmelse af konstanterne c_1 og c_2 så $\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 2 \\ 4 \end{bmatrix}$ haves det lneære ligningssystem

$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} c_1 - c_2 \\ c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 2 \\ 4 \end{bmatrix}$. Heraf fås $c_1 = 3$ og $c_2 = 1$.

den søgte flowkurve er da

```
> r:=t-><3-exp(t),3+exp(t)>:
> r(t)
```

$$\begin{bmatrix} 3 - e^t \\ 3 + e^t \end{bmatrix} \tag{3.1.3}$$

hvor $t \in \mathbb{R}$.

Kurven starter altså i punktet

```
> r(0)
```

$$\begin{bmatrix} 2 \\ 4 \end{bmatrix} \tag{3.1.4}$$

og går endvidere gennem punktet

```
> r(ln(3))
```

$$\begin{bmatrix} 0 \\ 6 \end{bmatrix} \tag{3.1.5}$$

En kurve K har parameterfremstillingen $\mathbf{s}(u) = \begin{bmatrix} u \\ u^2 \end{bmatrix}$, $u \in [0; 2]$.

Spørgsmål 2

Til bestemmelse af konstanterne c_1 og c_2 så $\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} u \\ u^2 \end{bmatrix}$ haves det lineære ligningssystem

$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} c_1 - c_2 \\ c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} u \\ u^2 \end{bmatrix}$ med tilhørende totalmatrix

```
> T:=<1,-1,u;1,1,u^2>
```

$$T := \begin{bmatrix} 1 & -1 & u \\ 1 & 1 & u^2 \end{bmatrix} \tag{3.2.1}$$

```
> c:=LinearSolve(T)
```

$$c := \begin{bmatrix} \frac{1}{2}\, u^2 + \frac{1}{2}\, u \\ \frac{1}{2}\, u^2 - \frac{1}{2}\, u \end{bmatrix} \tag{3.2.2}$$

Den flowkurve **r**(u, t), der opfylder **r**(u,0) = $\begin{bmatrix} u \\ u^2 \end{bmatrix}$ har da parameterfremstillingen

```
> r:=(u,t)-><c[1]-c[2]*exp(t),c[1]+c[2]*exp(t)>:
> r(u,t)
```

$$\begin{bmatrix} \frac{u^2}{2} + \frac{u}{2} - \left(\frac{1}{2}\, u^2 - \frac{1}{2}\, u\right) e^t \\ \frac{u^2}{2} + \frac{u}{2} + \left(\frac{1}{2}\, u^2 - \frac{1}{2}\, u\right) e^t \end{bmatrix} \tag{3.2.3}$$

For $t = \ln(3)$ og $u \in [0; 2]$ fås

```
> r(u,ln(3))
```

$$\begin{bmatrix} -u^2 + 2\, u \\ 2\, u^2 - u \end{bmatrix} \tag{3.2.4}$$

som er en parameterfremstilling for den kurve, som K er blevet omformet til ved tiden $t = \ln(3)$.

```
> p1:=plot([vop(r(u,ln(3))),u=0..2],scaling=constrained,color=red):
> p2:=plot([u,u^2,u=0..2],scaling=constrained,color=blue):
> p3:=pointplot([[2,4],[0,6]],symbol=solidcircle,color=[blue, red])
  :
> p4:=plot([3-exp(t),3+exp(t),t=0..ln(3)],linestyle=dot,color=
  darkgrey):
> display(p1,p2,p3,p4,labels=[x,y])
```



Opgaven kunne selvfølgelig også løses ved hjælp af dsolve.

Opgave 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)  end
  proc:
> with(LinearAlgebra):
```

Et vektorfelt **V** i (x, y, z)-rummet er givet ved

```
> V:=(x,y,z)-><-y*x,y^2+5,-y*z+5*z>:
> V(x,y,z)
```

$$\begin{bmatrix} -y\,x \\ y^2+5 \\ -y\,z+5\,z \end{bmatrix} \tag{4.1}$$

Vedrørende det givne massive prisme P med hjørnerne O, A, B, C, E og F henvises til figuren i opgaveteksten.

Spørgsmål 1

```
> divV:=div(V)(x,y,z)
```

$$divV := 5 \tag{4.1.1}$$

```
> rotV:=unapply(rot(V)(x,y,z),[x,y,z]):
> rotV(x,y,z)
```

$$\begin{bmatrix} -z \\ 0 \\ x \end{bmatrix} \tag{4.1.2}$$

$\mathrm{Vol}(P) = \frac{1}{2}\,\mathrm{Vol}(\mathrm{kasse}) = \frac{4}{2} = 2.$

Spørgsmål 2

∂P er den lukkede overflade af P orienteret med udadrettet enhedsnormalvektor. Af Gauss' sætning fås da

$$\mathrm{Flux}(\mathbf{V},\partial P) = \int_P \mathrm{Div}(\mathbf{V})\,\mathrm{d}\mu = \int_P 5\,\mathrm{d}\mu = 5\int_P \mathrm{d}\mu = 5\mathrm{Vol}(P) = 10.$$

Lad S betegne den sideflade af P, som har hjørnerne B, C, E og F.

Spørgsmål 3

Da S er rektanglet med hjørnerne B, C, E og F beliggende i planen med ligningen $z = 2-y$, så er en parameterfremstilling for S

```
> r:=(u,v)-><u,v,2-v>:
> r(u,v)
```

$$\begin{bmatrix} u \\ v \\ 2-v \end{bmatrix} \tag{4.3.1}$$

hvor $u \in [0; 1]$ og $v \in [0; 2]$.

```
> ru:=diff(r(u,v),u)
```

$$ru := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \tag{4.3.2}$$

```
> rv:=diff(r(u,v),v)
```

$$rv := \begin{bmatrix} 0 \\ 1 \\ -1 \end{bmatrix} \tag{4.3.3}$$

Fladens normalvektor

```
> N:=kryds(ru,rv)
```

$$N := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix} \tag{4.3.4}$$

danner netop højreskrue med den valgte orientering af den lukkede randkurve ∂S af S (vist med rød pil på figuren i opgaveteksten). Af Stokes' sætning fås da

$$\mathrm{Cirk}(\mathbf{V}, \partial S) = \int_{\partial S} \mathbf{V} \cdot \mathbf{e}_{\partial S}\, \mathrm{d}\mu = \mathrm{Flux}(\mathbf{Rot}(\mathbf{V}), S) = \int_S \mathbf{n}_S \cdot \mathbf{Rot}(\mathbf{V}) \mathrm{d}\mu = \int_0^2 \int_0^1 \mathbf{N}(u,v) \cdot \mathbf{Rot}(\mathbf{V})(\mathbf{r}(u.v)) \mathrm{d}u \mathrm{d}v$$

Rotationen taget på fladen

```
> Rot:=rotV(vop(r(u,v)))
```

$$Rot := \begin{bmatrix} -2 + v \\ 0 \\ u \end{bmatrix} \tag{4.3.5}$$

```
> integranden:=prik(Rot,N)
```

$$integranden := u \tag{4.3.6}$$

```
> Int(Int(integranden,u=0..1),v=0..2)=int(int(integranden,u=0..1),v=
  0..2)
```

$$\int_0^2 \left(\int_0^1 u \, \mathrm{d}u \right) \mathrm{d}v = 1 \tag{4.3.7}$$